
Changelog for v0.x
    


  
    
  
  v0.1.0


Requires Elixir v1.17+
Compatible with Ecto 3.2+

  
    
  
  Features


	Automatic CRUD function generation for Ecto schemas
	Customizable repository module selection
	Customizable changeset function names
	Support for both safe and raising versions of operations
	Built-in support for preloading associations



  

    
Overview
    

Aurora.Ctx is a code generation library that implements the Context pattern in Phoenix applications. It automatically generates standardized database interaction functions following Phoenix's architectural guidelines, while maintaining full compatibility with Ecto's query composition and changesets. The library supports both standard CRUD operations and advanced query features like filtering, sorting, and pagination.

  
    
  
  Key Benefits


	Reduced Boilerplate: Automatically generates common database access functions
	Consistent API: Ensures uniform function names and patterns across contexts
	Type Safety: All generated functions include proper typespecs
	Flexible: Supports customization through options and hooks
	Maintainable: Less code to maintain, centralized logic


  
    
  
  Getting Started


Add aurora_ctx to your list of dependencies in mix.exs:
def deps do
  [
    {:aurora_ctx, "~> 0.1.0"}
  ]
end

  
    
  
  Basic Usage



  
    
  
  Creating an Ecto Schema


Tip: Consider using TypedEctoSchema to reduce boilerplate in your schema definitions and add compile-time type checking.

Let's start with a typical Ecto schema definition that we'll use throughout the examples:
defmodule MyApp.Inventory.Product do
  use Ecto.Schema
  
  schema "products" do
    field :name, :string
    field :price, :decimal
    
    timestamps()
  end

  def changeset(product, attrs) do
    product
    |> cast(attrs, [:name, :price])
    |> validate_required([:name, :price])
  end
end

  
    
  
  Traditional Context Approach


Without Aurora.Ctx, you would typically write a context module with all the CRUD operations manually:
defmodule MyApp.Inventory do
  import Ecto.Query
  alias MyApp.Repo
  alias MyApp.Inventory.Product

  def list_products do
    Product |> from() |> Repo.all()
  end

  def get_product(id) do
    Product |> Repo.get(id)
  end

  def get_product!(id) do
    Product |> Repo.get!(id)
  end

  def create_product(attrs) do
    %Product{}
    |> Product.changeset(attrs)
    |> Repo.insert()
  end

  def update_product(%Product{} = product, attrs) do
    product
    |> Product.changeset(attrs)
    |> Repo.update()
  end

  def delete_product(%Product{} = product) do
    Repo.delete(product)
  end

  def change_product(%Product{} = product, attrs \\ %{}) do
    Product.changeset(product, attrs)
  end
end

  
    
  
  Context Generation with Aurora.Ctx


With Aurora.Ctx, the same functionality can be achieved with just a few lines of code:
defmodule MyApp.Inventory do
  use Aurora.Ctx

  # The repository will default to MyApp.Repo  
  ctx_register_schema(MyApp.Inventory.Product)
end
Now you can use the context just like the traditional approach:
# List all products
products = Inventory.list_products()

# Get a specific product
product = Inventory.get_product!(1)

# Create a new product
{:ok, product} = Inventory.create_product(%{name: "Widget", price: 19.99})

# Update a product
{:ok, product} = Inventory.update_product(product, %{price: 29.99})

# Delete a product
{:ok, _deleted} = Inventory.delete_product(product)

  
    
  
  Function Customization


You can override any generated function by defining it before calling ctx_register_schema:
defmodule MyApp.Inventory do
  use Aurora.Ctx

  def list_products do
    # Custom implementation with specific business logic
    MyApp.Inventory.Product
    |> where([p], p.active == true)
    |> order_by([p], [desc: p.inserted_at])
    |> MyApp.Repo.all()
  end

  ctx_register_schema(MyApp.Inventory.Product)
end

  
    
  
  Generated Functions


For each schema, the following functions are automatically generated:

  
    
  
  List Functions


	list_*()- List all records
	list_*(opts) - List all records with options
	list_*_paginated() - List records with pagination
	list_*_paginated(opts) - List records with pagination and options
	count_*() - Count total records
	count_*(opts) - Count records with options


  
    
  
  Create Functions


	create_*() - Create a record with empty attributes
	create_*(attrs) - Create a record with given attributes
	create_*!() - Create a record with empty attributes (raises on error)
	create_*!(attrs) - Create a record with given attributes (raises on error)


  
    
  
  Get Functions


	get_*(id) - Get a record by ID
	get_*(id, opts) - Get a record by ID with options
	get_*!(id) - Get a record by ID (raises if not found)
	get_*!(id, opts) - Get a record by ID with options (raises if not found)


  
    
  
  Delete Functions


	delete_*(entity) - Delete a record
	delete_*!(entity) - Delete a record (raises on error)


  
    
  
  Change Functions


	change_*(entity) - Create a changeset from a record
	change_*(entity, attrs) - Create a changeset from a record with attributes


  
    
  
  Update Functions


	update_*(entity) - Update a record
	update_*(entity, attrs) - Update a record with attributes


  
    
  
  New Functions


	new_*() - Initialize a new struct
	new_*(attrs) - Initialize a new struct with attributes
	new_*(attrs, opts) - Initialize a new struct with attributes and options

Check the Functions guide for detailed documentation of each function.

  
    
  
  Direct Core Usage


Sometimes you may want to use the core operations directly without generating context functions. This approach is particularly useful in scenarios such as:
	When working with dynamic schemas determined at runtime
	Building generic admin interfaces or APIs that handle multiple schemas
	Creating reusable modules that work with any schema
	Testing and prototyping before defining your final context structure

Here's how to use the core module directly:
alias Aurora.Ctx.Core

# List all records
products = Core.list(MyApp.Repo, MyApp.Inventory.Product)

# List with filtering and sorting
products = Core.list(MyApp.Repo, MyApp.Inventory.Product,
  where: [status: :active],
  order_by: [desc: :inserted_at],
  preload: [:category]
)

# Paginated listing
page = Core.list_paginated(MyApp.Repo, MyApp.Inventory.Product,
  paginate: %{page: 1, per_page: 20}
)

# Create a record
{:ok, product} = Core.create(MyApp.Repo, MyApp.Inventory.Product, %{
  name: "Widget",
  price: 19.99
})

# Update a record
{:ok, product} = Core.update(MyApp.Repo, product, %{price: 29.99})

# Delete a record
{:ok, _deleted} = Core.delete(MyApp.Repo, product)

  
    
  
  Context vs Direct Core Usage


While using the Core module directly provides flexibility, using proper contexts has several advantages:
	Encapsulation: Contexts hide implementation details and provide a cleaner API
	Business Logic: Contexts are the ideal place to add business rules and validations
	Consistency: Generated functions ensure consistent naming and behavior across your app
	Documentation: Function names in contexts are more descriptive of their business purpose
	Maintainability: Changes to database operations can be centralized in contexts
	Type Safety: Generated functions include proper typespecs for better tooling support

Consider using proper contexts for most application code, and reserve direct Core usage for special cases where flexibility is more important than the benefits listed above.


  

    
Generated Functions Reference
    

This guide details all the functions that Aurora.Ctx generates for your context modules.

  
    
  
  List Functions


list_products()                # Returns [%Product{}]
list_products(opts)            # Returns [%Product{}] with filtering/sorting
list_products_paginated()      # Returns %Pagination{} with default options
list_products_paginated(opts)  # Returns %Pagination{} with custom options
count_products()               # Returns total count of records
count_products(opts)           # Returns filtered count of records

  
    
  
  Example


# Get active products sorted by price
products = list_products(
  where: [status: :active],
  order_by: [asc: :price]
)
The opts parameter supports:
	:preload - Associations to preload
	:where - Filter conditions (equality, comparison, range)
	:or_where - Alternative filter conditions combined with OR
	:order_by - Sorting specification
	:paginate - Pagination options (page, per_page)


  
    
  
  Get Functions


get_product(id)                # Returns %Product{} or nil
get_product(id, opts)          # Returns %Product{} or nil with preloads
get_product!(id)               # Returns %Product{} or raises Ecto.NoResultsError
get_product!(id, opts)         # Returns %Product{} or raises, with preloads

  
    
  
  Example


# Get product with associated category
product = get_product!(1, preload: [:category])

  
    
  
  Create Functions


create_product()               # Returns {:ok, %Product{}} with defaults
create_product(attrs)          # Returns {:ok, %Product{}} or {:error, changeset}
create_product!()              # Returns %Product{} or raises
create_product!(attrs)         # Returns %Product{} or raises

  
    
  
  Example


{:ok, product} = create_product(%{
  name: "Widget Pro",
  price: Decimal.new("29.99")
})

  
    
  
  Update Functions


update_product(entity)         # Returns {:ok, %Product{}} with no changes
update_product(entity, attrs)  # Returns {:ok, %Product{}} or {:error, changeset}

  
    
  
  Example


{:ok, updated} = update_product(product, %{price: Decimal.new("39.99")})

  
    
  
  Delete Functions


delete_product(entity)         # Returns {:ok, %Product{}} or {:error, changeset}
delete_product!(entity)        # Returns %Product{} or raises

  
    
  
  Example


{:ok, deleted} = delete_product(product)

  
    
  
  Change Functions


change_product(entity)         # Returns %Ecto.Changeset{} with no changes
change_product(entity, attrs)  # Returns %Ecto.Changeset{} with changes

  
    
  
  Example


changeset = change_product(product, %{price: Decimal.new("49.99")})

  
    
  
  New Functions


new_product()                  # Returns %Product{} struct
new_product(attrs)             # Returns %Product{} with attributes
new_product(attrs, opts)       # Returns %Product{} with attributes and preloads

  
    
  
  Example


product = new_product(%{name: "Widget Pro", price: Decimal.new("29.99")})

  
    
  
  Query Options



  
    
  
  Where Conditions


# Simple equality
where: [status: :active]

# Comparisons
where: {:price, :greater_than, 100}
where: {:price, :less_than, 200}
where: {:date, :greater_equal_than, ~D[2023-01-01]}

# Ranges
where: {:price, :between, 100, 200}

# Multiple conditions
where: [status: :active, price: {:greater_than, 100}]

  
    
  
  Sorting


# Simple sort
order_by: :inserted_at           # ascending
order_by: {:desc, :inserted_at}  # descending

# Multiple fields
order_by: [{:desc, :inserted_at}, {:asc, :name}]

  
    
  
  Pagination


# Default pagination
paginate: %{page: 1, per_page: 20}
For implementation examples, check out the Examples guide.


  

    
Examples
    

Examples of using Aurora.Ctx in common scenarios.

  
    
  
  Basic CRUD Operations


Basic context setup and standard database operations.
defmodule MyApp.Inventory do
  use Aurora.Ctx,
    schema_module: MyApp.Inventory.Product,
    repo_module: MyApp.Repo
end

# Usage examples
alias MyApp.Inventory

# Create a product
{:ok, product} = Inventory.create_product(%{
  name: "Premium Widget",
  price: Decimal.new("29.99")
})

# List products with sorting
products = Inventory.list_products(order_by: [desc: :inserted_at])

# Get product with preloaded associations
{:ok, product} = Inventory.get_product(1, preload: [:category, :supplier])

# Update a product
{:ok, updated} = Inventory.update_product(product, %{price: Decimal.new("39.99")})

# Delete a product
{:ok, _deleted} = Inventory.delete_product(product)

  
    
  
  Query Usage


More complex database operations using query options.

  
    
  
  Custom Query Options


Combining filtering, preloading, and sorting in queries.
# List active products with preloaded categories
products = Inventory.list_products(
  where: [status: "active"],
  preload: [:category],
  order_by: [asc: :name],
  limit: 10
)

# Get a product with specific preloads
product = Inventory.get_product!(1, 
  preload: [
    category: [:parent_category],
    reviews: [:user]
  ]
)

  
    
  
  Working with Changesets


Data validation and form handling with changesets.
# Create a changeset for a new product
changeset = Inventory.change_product(%Product{})

# Create a changeset with attributes
changeset = Inventory.change_product(%Product{}, %{
  name: "New Widget",
  price: "19.99"
})

# Working with forms in Phoenix
def new(conn, _params) do
  changeset = Inventory.change_product(%Product{})
  render(conn, "new.html", changeset: changeset)
end

def create(conn, %{"product" => product_params}) do
  case Inventory.create_product(product_params) do
    {:ok, product} -> 
      redirect(to: Routes.product_path(conn, :show, product))
    {:error, changeset} ->
      render(conn, "new.html", changeset: changeset)
  end
end

  
    
  
  Query Options


Available options for customizing database queries.

  
    
  
  Filtering


Filter query results using different conditions.
# Multiple conditions
products = Inventory.list_products(
  where: [
    status: "active",
    category_id: 1
  ]
)

# Comparison operators
products = Inventory.list_products(
  where: [
    {:price, :gt, Decimal.new("100.00")},
    {:stock_level, :le, 5}
  ]
)

# Range queries
products = Inventory.list_products(
  where: {:price, :between, Decimal.new("10.00"), Decimal.new("50.00")}
)

# OR conditions
products = Inventory.list_products(
  where: [status: "active"],
  or_where: [status: "pending_review"]
)

  
    
  
  Sorting


Order query results with various sorting strategies.
# Multiple sort fields
products = Inventory.list_products(
  order_by: [
    asc: :category_id,
    desc: :price,
    asc: :name
  ]
)

# Null handling
products = Inventory.list_products(
  order_by: [
    asc_nulls_last: :discontinued_at,
    desc_nulls_first: :updated_at
  ]
)

  
    
  
  Controlled Pagination


Implement and navigate through paginated results. Results are wrapped in an Aurora.Ctx.Pagination struct that provides additional metadata and navigation helpers.
# Basic pagination
page1 = Inventory.list_products_paginated(
  paginate: %{page: 1, per_page: 20}
)

# Access pagination info
IO.puts "Total entries: #{page1.entries_count}"
IO.puts "Total pages: #{page1.pages_count}"
IO.puts "Current page: #{page1.page}"

# Navigate through pages
next_page = Aurora.Ctx.Core.next_page(page1)
prev_page = Aurora.Ctx.Core.previous_page(next_page)
page5 = Aurora.Ctx.Core.to_page(page1, 5)

# Combine with other options
filtered_page = Inventory.list_products_paginated(
  where: [category_id: 1],
  order_by: [desc: :inserted_at],
  paginate: %{page: 1, per_page: 10}
)

  
    
  
  Preloading Associates


Load associated data with simple and nested preloads.
# Nested preloads
product = Inventory.get_product!(1,
  preload: [
    category: [:parent_category, :subcategories],
    reviews: [:user, comments: [:user]],
    supplier: [:address, :contacts]
  ]
)

# Filtered preloads with custom queries
products = Inventory.list_products(
  preload: [
    reviews: from(r in Review, where: r.rating > 3)
  ]
)

  
    
  
  Working with Transactions


Combine multiple database operations in transactions.
alias Ecto.Multi
alias MyApp.{Inventory, Orders}

# Complex multi-operation transaction
def create_order_with_inventory_update(attrs) do
  Multi.new()
  |> Multi.run(:product, fn repo, _ ->
    Inventory.get_product(attrs.product_id)
  end)
  |> Multi.run(:check_stock, fn _repo, %{product: product} ->
    if product.stock >= attrs.quantity do
      {:ok, product}
    else
      {:error, :insufficient_stock}
    end
  end)
  |> Multi.run(:order, fn repo, %{product: product} ->
    Orders.create_order(attrs)
  end)
  |> Multi.run(:update_inventory, fn repo, %{product: product, order: order} ->
    Inventory.update_product(product, %{
      stock: product.stock - order.quantity
    })
  end)
  |> MyApp.Repo.transaction()
end

# Usage
case create_order_with_inventory_update(%{product_id: 1, quantity: 5}) do
  {:ok, %{order: order, product: product}} ->
    {:ok, order}
  {:error, failed_operation, failed_value, changes_so_far} ->
    {:error, :order_creation_failed}
end

  
    
  
  Custom Changesets


Configure and use custom changeset functions.
# Using custom changeset functions
defmodule MyApp.Inventory do
  use Aurora.Ctx,
    schema_module: MyApp.Inventory.Product,
    repo_module: MyApp.Repo,
    changeset_function: :custom_changeset,
    create_changeset_function: :create_changeset

  # Your additional context functions...
end

# Schema with multiple changeset functions
defmodule MyApp.Inventory.Product do
  use Ecto.Schema
  import Ecto.Changeset

  schema "products" do
    field :name, :string
    field :price, :decimal
    field :status, :string
    timestamps()
  end

  def create_changeset(product, attrs) do
    product
    |> cast(attrs, [:name, :price])
    |> validate_required([:name, :price])
    |> validate_number(:price, greater_than: 0)
  end

  def custom_changeset(product, attrs) do
    product
    |> cast(attrs, [:name, :price, :status])
    |> validate_required([:status])
    |> validate_inclusion(:status, ["active", "inactive"])
  end
end

# Usage
product = Inventory.new_product()
changeset = Inventory.change_product(product, %{name: "New Product"})
For more examples and use cases, check the test files in the GitHub repository.


  

    
Aurora.Ctx.Core 
    



      
Core implementation of common database operations using Ecto.
Provides functions for:
	CRUD operations (create, read, update, delete)
	Pagination (list_paginated, navigation)
	Query operations (count, list)
	Record management (change, new)


      


      
        Summary


  
    Functions
  


    
      
        change(entity, attrs)

      


        Creates a changeset for the given entity.



    


    
      
        change(changeset, changeset_function \\ :changeset, attrs \\ %{})

      


        Creates a changeset for the given entity using a specific changeset function.



    


    
      
        count(repo_module, schema_module, opts \\ [])

      


        Counts total records matching query conditions.



    


    
      
        create(repo_module, schema_module, attrs)

      


        Creates a new record.



    


    
      
        create(repo_module, schema_module, changeset_function \\ :changeset, attrs \\ %{})

      


        Creates a new record with optional custom changeset function.



    


    
      
        create!(repo_module, schema_module, attrs)

      


        Creates a new record, raising on errors.



    


    
      
        create!(repo_module, schema_module, changeset_function \\ :changeset, attrs \\ %{})

      


        Creates a new record, raising on errors.



    


    
      
        delete(repo_module, entity)

      


        Deletes the given entity.



    


    
      
        delete!(repo_module, entity)

      


        Deletes the given entity, raising on errors.



    


    
      
        exclude_clauses(query, clauses)

      


        Excludes query clauses from an Ecto query.



    


    
      
        get(repo_module, schema_module, id, opts \\ [])

      


        Gets a record by id.



    


    
      
        get!(repo_module, schema_module, id, opts \\ [])

      


        Gets a record by id, raising if not found.



    


    
      
        list(repo_module, schema_module, opts \\ [])

      


        Lists all records for a schema with optional filtering and sorting.



    


    
      
        list_paginated(repo_module, schema_module, opts \\ [])

      


        Lists records with pagination support.



    


    
      
        new(repo_module, schema_module, attrs)

      


        Initializes a new schema struct with optional attributes and preloads.



    


    
      
        new(repo_module, schema_module, attrs \\ %{}, opts \\ [])

      


        Initializes a new struct with optional preloads.



    


    
      
        next_page(paginate)

      


        Moves to next page in paginated results.



    


    
      
        previous_page(paginate)

      


        Moves to previous page in paginated results.



    


    
      
        to_page(paginate, page)

      


        Changes to a specific page in paginated results.



    


    
      
        update(repo_module, entity, attrs)

      


        Updates an entity with given attributes.



    


    
      
        update(repo_module, entity, changeset_function \\ :changeset, attrs \\ %{})

      


        Updates an entity using a specific changeset function.



    





      


      
        Functions

        


  
    
      
    
    
      change(entity, attrs)



        
          
        

    

  


  

      

          @spec change(Ecto.Schema.t() | Ecto.Changeset.t(), map() | nil) :: Ecto.Changeset.t()


      


Creates a changeset for the given entity.
Parameters:
	entity (Ecto.Schema.t()) - Entity for changeset
	attrs (map) - Changeset attributes

Returns Ecto.Changeset.

  



    

  
    
      
    
    
      change(changeset, changeset_function \\ :changeset, attrs \\ %{})



        
          
        

    

  


  

      

          @spec change(Ecto.Schema.t() | Ecto.Changeset.t(), atom(), map() | nil) ::
  Ecto.Changeset.t()


      


Creates a changeset for the given entity using a specific changeset function.

  
    
  
  Parameters


	entity (Ecto.Schema.t() | Ecto.Changeset.t()) - Existing record or changeset

	changeset_function (atom()) - Custom changeset function
	attrs (map()) - Changeset attributes


  
    
  
  Returns


	Ecto.Changeset.t()


  
    
  
  Examples


alias MyApp.Product
product = %Product{name: "Widget"}
Core.change(product, %{name: "New Widget"})
#=> #Ecto.Changeset<changes: %{name: "New Widget"}>

Core.change(product, :custom_changeset, %{status: "active"})
#=> #Ecto.Changeset<changes: %{status: "active"}>

  



    

  
    
      
    
    
      count(repo_module, schema_module, opts \\ [])



        
          
        

    

  


  

      

          @spec count(module(), module(), keyword()) :: non_neg_integer()


      


Counts total records matching query conditions.
Parameters:
	repo_module (module) - Ecto.Repo to use
	schema_module (module) - Schema to query
	opts (keyword) - Optional query filtering options

Returns total count of matching records.

  



  
    
      
    
    
      create(repo_module, schema_module, attrs)



        
          
        

    

  


  

      

          @spec create(module(), module(), map() | nil) ::
  {:ok, Ecto.Schema.t()} | {:error, Ecto.Changeset.t()}


      


Creates a new record.
Parameters:
	repo_module (module) - Ecto.Repo to use
	schema_module (module) - Schema to create
	attrs (map) - Attributes for the new record

Returns {:ok, schema} on success, {:error, changeset} on failure.

  



    

  
    
      
    
    
      create(repo_module, schema_module, changeset_function \\ :changeset, attrs \\ %{})



        
          
        

    

  


  

      

          @spec create(module(), module(), atom(), map() | nil) ::
  {:ok, Ecto.Schema.t()} | {:error, Ecto.Changeset.t()}


      


Creates a new record with optional custom changeset function.

  
    
  
  Parameters


	repo_module (module()) - Ecto.Repo module to use
	schema_module (module()) - Schema module for the record
	changeset_function (atom()) - Custom changeset function to use
	attrs (map()) - Attributes for the new record


  
    
  
  Returns


	{:ok, Ecto.Schema.t()} - On success
	{:error, Ecto.Changeset.t()} - On validation failure


  
    
  
  Examples


alias MyApp.{Repo, Product}
Core.create(Repo, Product, %{name: "Widget"})
#=> {:ok, %Product{name: "Widget"}}

Core.create(Repo, Product, :custom_changeset, %{
  name: "Widget",
  status: "active"
})
#=> {:ok, %Product{name: "Widget", status: "active"}}

  



  
    
      
    
    
      create!(repo_module, schema_module, attrs)



        
          
        

    

  


  

      

          @spec create!(module(), module(), map() | nil) :: Ecto.Schema.t()


      


Creates a new record, raising on errors.
Parameters:
	repo_module (module) - Ecto.Repo to use
	schema_module (module) - Schema to create
	attrs (map) - Attributes for the new record

Returns created schema or raises on error.

  



    

  
    
      
    
    
      create!(repo_module, schema_module, changeset_function \\ :changeset, attrs \\ %{})



        
          
        

    

  


  

      

          @spec create!(module(), module(), atom(), map() | nil) :: Ecto.Schema.t()


      


Creates a new record, raising on errors.
Parameters:
	repo_module (module) - Ecto.Repo to use
	schema_module (module) - Schema to create
	changeset_function (atom) - Changeset function to use
	attrs (map) - Attributes for the new record

Returns created schema or raises on error.

  



  
    
      
    
    
      delete(repo_module, entity)



        
          
        

    

  


  

      

          @spec delete(module(), Ecto.Schema.t()) ::
  {:ok, Ecto.Schema.t()} | {:error, Ecto.Changeset.t()}


      


Deletes the given entity.
Parameters:
	repo_module (module): The Ecto.Repo module to use
	entity (Ecto.Schema.t()): The entity to delete

Returns: {:ok, Ecto.Schema.t()} | {:error, Ecto.Changeset.t()}

  



  
    
      
    
    
      delete!(repo_module, entity)



        
          
        

    

  


  

      

          @spec delete!(module(), Ecto.Schema.t()) :: Ecto.Schema.t()


      


Deletes the given entity, raising on errors.
Parameters:
	repo_module (module): The Ecto.Repo module to use
	entity (Ecto.Schema.t()): The entity to delete

Returns: Ecto.Schema.t()
Raises: Ecto.StaleEntryError

  



  
    
      
    
    
      exclude_clauses(query, clauses)



        
          
        

    

  


  

      

          @spec exclude_clauses(Ecto.Query.t(), atom() | [atom()]) :: Ecto.Query.t()


      


Excludes query clauses from an Ecto query.
Parameters:
	query (Ecto.Query.t()) - Query to modify
	clauses (atom | [atom]) - Clause(s) to exclude


Returns modified query.

  



    

  
    
      
    
    
      get(repo_module, schema_module, id, opts \\ [])



        
          
        

    

  


  

      

          @spec get(module(), module(), term(), keyword()) :: Ecto.Schema.t() | nil


      


Gets a record by id.
Parameters:
	repo_module (module) - Ecto.Repo to use
	schema_module (module) - Schema to query
	id (term) - Primary key to look up
	opts (keyword) - Optional query parameters

Returns found schema or nil if not found.

  



    

  
    
      
    
    
      get!(repo_module, schema_module, id, opts \\ [])



        
          
        

    

  


  

      

          @spec get!(module(), module(), term(), keyword()) :: Ecto.Schema.t()


      


Gets a record by id, raising if not found.
Parameters:
	repo_module (module) - Ecto.Repo to use
	schema_module (module) - Schema to query
	id (term) - Primary key to look up
	opts (keyword) - Optional query parameters

Returns found schema or raises Ecto.NoResultsError.

  



    

  
    
      
    
    
      list(repo_module, schema_module, opts \\ [])



        
          
        

    

  


  

      

          @spec list(module(), module(), keyword()) :: list()


      


Lists all records for a schema with optional filtering and sorting.

  
    
  
  Parameters


	repo_module (module()) - Ecto.Repo module to use
	schema_module (module()) - Schema module to query
	opts (keyword()) - Optional query parameters:	:where - Filter conditions
	:order_by - Sort specifications
	:preload - Associations to preload
	:limit - Maximum number of records




  
    
  
  Returns


	list(Ecto.Schema.t())


  
    
  
  Examples


alias MyApp.{Repo, Product}
Core.list(Repo, Product)
#=> [%Product{}, ...]

Core.list(Repo, Product,
  where: [status: :active],
  order_by: [desc: :inserted_at],
  preload: [:category]
)
#=> [%Product{category: %Category{}}, ...]

  



    

  
    
      
    
    
      list_paginated(repo_module, schema_module, opts \\ [])



        
          
        

    

  


  

      

          @spec list_paginated(module(), module(), keyword()) :: Aurora.Ctx.Pagination.t()


      


Lists records with pagination support.

  
    
  
  Parameters


	repo_module (module()) - Ecto.Repo module to use
	schema_module (module()) - Schema module to query
	opts (keyword()) - Query options including:	:paginate - Pagination settings (%{page: integer, per_page: integer})
	:where - Filter conditions
	:order_by - Sort specifications
	:preload - Associations to preload




  
    
  
  Configuration


Default pagination settings can be configured in your config.exs:
config :aurora_ctx, :pagination,
  page: 1,
  per_page: 40
If not configured, defaults to page: 1, per_page: 40.

  
    
  
  Returns


	Aurora.Ctx.Pagination.t()


  
    
  
  Examples


alias MyApp.{Repo, Product}
Core.list_paginated(Repo, Product,
  paginate: %{page: 1, per_page: 20},
  where: [category_id: 1],
  order_by: [desc: :inserted_at]
)
#=> %Aurora.Ctx.Pagination{
#=>   entries: [%Product{}],
#=>   page: 1,
#=>   per_page: 20,
#=>   entries_count: 50,
#=>   pages_count: 3
#=> }

  



  
    
      
    
    
      new(repo_module, schema_module, attrs)



        
          
        

    

  


  

      

          @spec new(module(), module(), map() | keyword()) :: Ecto.Schema.t()


      


Initializes a new schema struct with optional attributes and preloads.
Parameters:
	repo_module (module) - Ecto.Repo to use
	schema_module (module) - Schema to initialize
	attrs (map) - Initial attributes
	opts (keyword) - Optional preload parameters

Returns initialized schema struct.

  



    

  
    
      
    
    
      new(repo_module, schema_module, attrs \\ %{}, opts \\ [])



        
          
        

    

  


  

      

          @spec new(module(), module(), map(), keyword()) :: Ecto.Schema.t()


      


Initializes a new struct with optional preloads.
Parameters:
	repo_module (module): The Ecto.Repo module to use
	schema_module (module): The Ecto.Schema module to initialize
	opts (keyword): Optional preload parameters

Returns: Ecto.Schema.t()

  



  
    
      
    
    
      next_page(paginate)



        
          
        

    

  


  

      

          @spec next_page(Aurora.Ctx.Pagination.t() | map()) :: Aurora.Ctx.Pagination.t()


      


Moves to next page in paginated results.
Parameters:
	paginate (Pagination.t()) - Current pagination state

Returns updated Pagination struct with next page entries.

  



  
    
      
    
    
      previous_page(paginate)



        
          
        

    

  


  

      

          @spec previous_page(Aurora.Ctx.Pagination.t() | map()) :: Aurora.Ctx.Pagination.t()


      


Moves to previous page in paginated results.
Parameters:
	paginate (Pagination.t()) - Current pagination state

Returns updated Pagination struct with previous page entries.

  



  
    
      
    
    
      to_page(paginate, page)



        
          
        

    

  


  

      

          @spec to_page(Aurora.Ctx.Pagination.t() | map(), integer()) ::
  Aurora.Ctx.Pagination.t()


      


Changes to a specific page in paginated results.
Parameters:
	paginate (Pagination.t()) - Current pagination state
	page (integer) - Target page number

Returns updated Pagination struct with new page entries.

  



  
    
      
    
    
      update(repo_module, entity, attrs)



        
          
        

    

  


  

      

          @spec update(module(), Ecto.Schema.t(), map() | nil) ::
  {:ok, Ecto.Schema.t()} | {:error, Ecto.Changeset.t()}


      


Updates an entity with given attributes.
Parameters:
	repo_module (module) - Ecto.Repo to use
	entity (Ecto.Schema.t()) - Entity to update
	attrs (map) - Update attributes

Returns {:ok, schema} on success, {:error, changeset} on failure.

  



    

  
    
      
    
    
      update(repo_module, entity, changeset_function \\ :changeset, attrs \\ %{})



        
          
        

    

  


  

      

          @spec update(module(), Ecto.Schema.t(), atom(), map() | nil) ::
  {:ok, Ecto.Schema.t()} | {:error, Ecto.Changeset.t()}


      


Updates an entity using a specific changeset function.

  
    
  
  Parameters


	repo_module (module()) - Ecto.Repo module to use
	entity (Ecto.Schema.t()) - Existing record to update
	changeset_function (atom()) - Custom changeset function
	attrs (map()) - Update attributes


  
    
  
  Returns


	{:ok, Ecto.Schema.t()} - On success
	{:error, Ecto.Changeset.t()} - On validation failure


  
    
  
  Examples


alias MyApp.{Repo, Product}
product = %Product{name: "Old Name"}
Core.update(Repo, product, %{name: "New Name"})
#=> {:ok, %Product{name: "New Name"}}

  


        

      


  

    
Aurora.Ctx.Pagination 
    



      
Struct for handling pagination state.

  
    
  
  Configuration


Default pagination settings can be configured in your config.exs:
config :aurora_ctx, :pagination,
  page: 1,
  per_page: 40
If not configured, defaults to page: 1, per_page: 40.

      


      
        Summary


  
    Types
  


    
      
        t()

      


    





  
    Functions
  


    
      
        new()

      


        Creates a new pagination struct with safe defaults.



    


    
      
        new(attrs)

      


        Creates a new pagination struct with safe defaults.



    





      


      
        Types

        


  
    
      
    
    
      t()



        
          
        

    

  


  

      

          @type t() :: %Aurora.Ctx.Pagination{
  entries: list(),
  entries_count: non_neg_integer() | nil,
  opts: keyword(),
  page: pos_integer(),
  pages_count: pos_integer() | nil,
  per_page: pos_integer(),
  repo_module: module() | nil,
  schema_module: module() | nil
}


      



  


        

      

      
        Functions

        


  
    
      
    
    
      new()



        
          
        

    

  


  

      

          @spec new() :: t()


      


Creates a new pagination struct with safe defaults.
Returns:
  t()

  



  
    
      
    
    
      new(attrs)



        
          
        

    

  


  

      

          @spec new(map() | keyword()) :: t()


      


Creates a new pagination struct with safe defaults.
Parameters:
	attrs (map | keyword): Pagination attributes
	:page (pos_integer): Page number (default: 1)
	:per_page (pos_integer): Items per page (default: 40)



Returns:
  t()

  


        

      


  

    
Aurora.Ctx 
    



      
A module for automatically generating CRUD operations for Ecto schemas.
This module provides macros to generate standard database operation functions
based on Ecto schemas. It simplifies the creation of context modules by
automatically implementing common CRUD operations.

  
    
  
  Usage


defmodule MyApp.Accounts do
  use Aurora.Ctx

  ctx_register_schema(User)
  # or with options
  ctx_register_schema(User, CustomRepo,
    changeset_function: :custom_changeset,
    create_changeset_function: :create_changeset
  )
end

  
    
  
  Generated Functions


For a schema with source name "users" and module name "user", the following functions are generated:

  
    
  
  List Functions


	list_users/0 - List all records
	list_users/1 - List all records with options
	list_users_paginated/0 - List records with pagination
	list_users_paginated/1 - List records with pagination and options
	count_users/0 - Count total records
	count_users/1 - Count records with options


  
    
  
  Create Functions


	create_user/0 - Create a record with empty attributes
	create_user/1 - Create a record with given attributes
	create_user!/0 - Create a record with empty attributes (raises on error)
	create_user!/1 - Create a record with given attributes (raises on error)


  
    
  
  Get Functions


	get_user/1 - Get a record by ID
	get_user/2 - Get a record by ID with options
	get_user!/1 - Get a record by ID (raises if not found)
	get_user!/2 - Get a record by ID with options (raises if not found)


  
    
  
  Delete Functions


	delete_user/1 - Delete a record
	delete_user!/1 - Delete a record (raises on error)


  
    
  
  Change Functions


	change_user/1 - Create a changeset from a record
	change_user/2 - Create a changeset from a record with attributes


  
    
  
  Update Functions


	update_user/1 - Update a record
	update_user/2 - Update a record with attributes


  
    
  
  New Functions


	new_user/0 - Initialize a new struct
	new_user/1 - Initialize a new struct with attributes
	new_user/2 - Initialize a new struct with attributes and options


      


      
        Summary


  
    Functions
  


    
      
        ctx_register_schema(schema_module, repo \\ nil, opts \\ [])

      


        Registers a schema for CRUD function generation.



    


    
      
        implementable_functions(schema_module)

      


        Lists all implementable CRUD functions for a given schema.



    





      


      
        Functions

        


    

    

  
    
      
    
    
      ctx_register_schema(schema_module, repo \\ nil, opts \\ [])


        (macro)


        
          
        

    

  


  

Registers a schema for CRUD function generation.

  
    
  
  Parameters


	schema_module (module()) - The Ecto schema module to generate functions for
	repo (module() | nil) - (Optional) The Ecto.Repo to use for database operations. Can be set globally with @ctx_repo_module

	opts (keyword()) - (Optional) Configuration options:	:changeset_function (atom()) - The function to use for changesets (default: :changeset)
	:create_changeset_function (atom()) - A specific function to use for creation changesets




  
    
  
  Repository Configuration


The Ecto.Repo can be configured in two ways:
	Passing it as the second argument to ctx_register_schema/3
	Setting the @ctx_repo_module module attribute in your context

If neither is specified, it will attempt to use YourApp.Repo based on your
context module's namespace.

  
    
  
  Example


# Using explicit repo
ctx_register_schema(User, MyApp.Repo)

# Using module attribute
@ctx_repo_module MyApp.Repo
ctx_register_schema(User)

# With additional options
ctx_register_schema(User, MyApp.Repo, changeset_function: :custom_changeset)

  



  
    
      
    
    
      implementable_functions(schema_module)



        
          
        

    

  


  

      

          @spec implementable_functions(module()) :: list()


      


Lists all implementable CRUD functions for a given schema.
This function is used internally to determine which functions should be generated
for a given schema. It returns a list of maps containing function definitions.

  
    
  
  Parameters


	schema_module - The Ecto schema module to analyze


  
    
  
  Returns


	List of maps containing:	:type - The function type (:list, :create, etc.)
	:name - The function name
	:arity - The function arity




  


        

      


  

    
Aurora.Ctx.QueryBuilder 
    



      
Provides functions for building and composing Ecto queries with common operations like filtering,
sorting, pagination, and preloading associations.

      


      
        Summary


  
    Functions
  


    
      
        options(query, options \\ [])

      


        Applies a list of query options to an Ecto query.



    





      


      
        Functions

        


    

  
    
      
    
    
      options(query, options \\ [])



        
          
        

    

  


  

      

          @spec options(
  Ecto.Query.t() | nil,
  keyword()
) :: Ecto.Query.t()


      


Applies a list of query options to an Ecto query.

  
    
  
  Parameters


	query (Ecto.Query.t() | nil) - Core query to modify

	options (keyword) - List of query options to apply


  
    
  
  Options


	:preload (atom | list) - Associations to preload

	:where (keyword | map | tuple) - Filter conditions
	Simple equality: {:field, value}
	Comparison: {:field, operator, value} where operator can be:	:greater_than, :gt - Greater than
	:greater_equal_than, :ge - Greater than or equal
	:less_than, :lt - Less than
	:less_equal_than, :le - Less than or equal
	:equal_to, :eq - Equal to


	Range: {:field, :between, start_value, end_value}


	:or_where (keyword | map | tuple) - Same as :where but combines with OR

	:paginate (map) - Pagination options with keys:	:page (integer) - Page number
	:per_page (integer) - Items per page


	:sort (atom | tuple | list) - Sorting options:
	field (atom | {:asc | :desc, field}) - Field to sort by ascending

	fields ([{:asc | :desc, field}]) - List of sort fields specifications





  
    
  
  Returns


	Ecto.Query.t() - Modified query with all options applied


  
    
  
  Examples


query = from(p in Product)
QueryBuilder.options(query,
  where: [status: :active],
  preload: [:category],
  sort: [desc: :inserted_at],
  paginate: %{page: 1, per_page: 20}
)

  


        

      


  

    
mix test.setup 
    



      
A Mix task for setting up the test environment by creating and migrating the test database.

      


      
        Summary


  
    Functions
  


    
      
        run(args)

      


        Runs the test setup task by creating and migrating the test database.



    





      


      
        Functions

        


  
    
      
    
    
      run(args)



        
          
        

    

  


  

      

          @spec run(list() | nil) :: any()


      


Runs the test setup task by creating and migrating the test database.
Parameters:
	args: list | nil - Command line arguments (not used)


Returns:
	any

